Theory of Elasticity, Article 21, Cantilever

This example, shown in Figure 1, is found in the third edition of Theory of Elasticity by Timoshenko & Goodier, published in 1969 by McGraw-Hill. It is essentially an application of 2D continuum elasticity theory using stress functions. The notation and axis directions in this document are somewhat different from in the book. The unit-width cantilever in Figure 1 is considered, where the fixed support is placed at the right-hand side for convenience when using stress functions.



Figure 1: The beam considered in this example.

Input values in N and mm

```
input1 = {h → 1000, E → 200 000, ν → 0.3, P → -10 000 000, L → 1500};
```

$$\text{input2} = \left\{G \to \frac{E}{2\,(1+\nu)},\ I \to \frac{h^3}{12}\right\};$$

Stress function

The stress function used in this example is a combination of a second-order term that implies pure shear and a fourth-order term:

$$\Phi = C_{xz}\, x\, z + C_{xz3}\, x\, z^3;$$

Individually they both satisfy the fourth-order differentiatial equation. In terms of symbolic expressions, the stress function gives these stresses:

$\sigma_{xx} = \partial_z \partial_z \Phi$

which yields: $6\ x\ z\ C_{xz3}$

$\sigma_{zz} = \partial_x \partial_x \Phi$

which yields: 0

$\tau_{xz} = -\partial_x \partial_z \Phi$

which yields: $-C_{xz} - 3\ z^2\ C_{xz3}$

Boundary conditions

The two unknown coefficients C_{xz} and C_{xz3} are determined by the two boundary conditions on the stress (we later need further conditions to determine the displacement):

- Zero shear stress at the top and bottom surfaces of the beam: $\tau_{xz}\left(z = \pm \frac{h}{2}\right) = 0$
- Shear stress with sum $-P$ at the free end (negative because positive τ_{xz} acts downwards): $\int_{-\frac{h}{2}}^{\frac{h}{2}} \tau_{xz}(x = 0)\, dz = -P$

Solving those two equations:

$$\text{solution} = \text{Solve}\left[\left\{\left(\tau_{xz}\ /.\ z \to \frac{h}{2}\right) == 0\ \&\&\ \left(\left(\int_{-\frac{h}{2}}^{\frac{h}{2}} \tau_{xz}\, dz\right)\ /.\ x \to 0\right) == -P\right\},\ \{C_{xz},\ C_{xz3}\}\right]$$

which yields: $\left\{\left\{C_{xz} \to \frac{3\ P}{2\ h},\ C_{xz3} \to -\frac{2\ P}{h^3}\right\}\right\}$

Updating the stress expressions:

$\sigma_{xx} = \partial_z \partial_z \Phi\ /.\ \text{solution}$

which yields: $\left\{-\frac{12\ P\ x\ z}{h^3}\right\}$

$\sigma_{zz} = \partial_x \partial_x \Phi\ /.\ \text{solution}$

which yields: $\{0\}$

```
τxz = -∂x ∂z Φ /. solution
```

which yields: $\left\{-\frac{3\,P}{2\,h}+\frac{6\,P\,z^2}{h^3}\right\}$

Plot of stresses

```
σxxPlot = Plot3D[σxx /. input2 /. input1, {x, 0, L /. input1},
   {z, -h/2 /. input1, h/2 /. input1}, AxesLabel → {x, z}, PlotLabel → σxx,
   Ticks → {{}, {}, {0}}];
τxzPlot = Plot3D[τxz /. input2 /. input1, {x, 0, L /. input1},
  {z, -h/2 /. input1, h/2 /. input1}, AxesLabel → {x, z}, Ticks → {{}, {}, {0}},
  PlotLabel → τxz];
GraphicsGrid[{{σxxPlot, τxzPlot}}]
```

which yields:



Comparison of stresses with Euler-Bernoully beam theory

The solution presented above gives the following maximum axial stress at the outermost fibre at the fixed support, which is in compression (negative) when the load *P* is positive (upwards):

```
σxx /. solution /. {x → L, z → h/2}
```

which yields: $\left\{\left\{-\frac{6\,L\,P}{h^2}\right\}\right\}$

Remembering that we here consider a unit width, b=1, and that the moment of inertia is $I = \frac{bh^3}{12}$the expression from beam theory becomes:

$$\sigma_{xx} = \frac{M}{I} z = \frac{P\,L}{\frac{h^3}{12}} \frac{h}{2} \tag{1}$$

which simplifies to the same expression as above:

```
(P L)/(h^3/12) h/2 // Simplify
```

which yields: $\frac{6\,L\,P}{h^2}$

The solution presented above gives the following maximum shear stress at the neutral axis (negative because P is positive upwards on the left-hand side, where positive shear stress is downwards):

```
τxz /. solution /. {x → 0, z → 0 }
```

which yields: $\left\{\left\{-\frac{3\,P}{2\,h}\right\}\right\}$

Again remembering that we are considering a beam of unit width the expression for maximum shear stress from beam theory is the same:

$$\tau_{xz} = \frac{3}{2}\frac{V}{A} = \frac{3}{2}\frac{P}{b\,h} = \frac{3}{2}\frac{P}{h} \tag{2}$$

Displacements

Displacements are obtained from integrating strains, and the strains are obtained from the stresses. The relevant strains are, according to the plane stress material law:

$$\epsilon_{xx} = \frac{\sigma_{xx}}{E} - \nu \frac{\sigma_{zz}}{E}$$

which yields: $\left\{-\frac{12\,P\,x\,z}{h^3\,E}\right\}$

$$\epsilon_{zz} = \frac{\sigma_{zz}}{E} - \nu \frac{\sigma_{xx}}{E}$$

which yields: $\left\{\frac{12\,P\,x\,z\,\nu}{h^3\,E}\right\}$

$$\gamma_{xz} = \frac{\tau_{xz}}{G} \text{ // Simplify}$$

which yields: $\left\{-\frac{3\,P\,(h^2 - 4\,z^2)}{2\,G\,h^3}\right\}$

The kinematic compatibility equations that relate the strains to displacements are:

$$\epsilon_{xx} = \frac{\partial u}{\partial x} \tag{3}$$

$$\epsilon_{zz} = \frac{\partial w}{\partial z} \tag{4}$$

$$\gamma_{xz} = \frac{\partial u}{\partial z} + \frac{\partial w}{\partial x} \tag{5}$$

Integration of Eq. (3) adds an integration constant that may vary with *z*, which is determined below so we clear it here in case the spreadsheet is recalculated:

```
Clear[Kz]
```

$$u = \int \epsilon_{xx}\, dx + Kz$$

which yields: $\left\{Kz - \frac{6\,P\,x^2\,z}{h^3\,E}\right\}$

Integration of Eq. (4) adds an integration constant that may vary with *x*:

$$\texttt{Clear[Kx]}$$

$$w = \int \epsilon_{zz}\, dz + Kx$$

which yields: $\left\{Kx + \frac{6\, P\, x\, z^2\, \nu}{h^3\, E}\right\}$

Those expressions for *u* and *v* are then substituted into Eq. (5), where we must manually differentiate Kx and Kz because *Mathematica* does not understand that they are not constants:

$$\gamma_{xz} == \partial_z u + dzKz + \partial_x w + dxKx \;//\; \text{Expand}$$

which yields: $\left\{-\frac{3\, P}{2\, G\, h} + \frac{6\, P\, z^2}{G\, h^3}\right\} == \left\{dxKx + dzKz - \frac{6\, P\, x^2}{h^3\, E} + \frac{6\, P\, z^2\, \nu}{h^3\, E}\right\}$

As explained on Page 43 in the book by Timoshenko the terms that vary with *x* must be constant, as must the terms with *z*, because *x* and *z* vary independently. This gives two equations that define those two constants (this is here done manually, not by *Mathematica*):

$$\frac{6\, P\, x^2}{h^3\, E} - dxKx == \text{Constant1};$$

$$\frac{6\, P\, z^2}{G\, h^3} - \frac{6\, P\, z^2\, \nu}{h^3\, E} - dzKz == \text{Constant2};$$

Using these definitions, the previous expression for γ_{xz} then becomes:

$$\text{Condition1} = \text{Constant1} + \text{Constant2} == \frac{3\, P}{2\, G\, h};$$

The two expressions that defined Constant1 and Constant2 are used to determine expressions for Kx and Kz (this is here done manually, not by *Mathematica*):

$$Kx = \frac{6\, P\, x^3}{3\, h^3\, E} - x\, \text{Constant1} + \text{Constant3};$$

$$Kz = \frac{6\, P\, z^3}{3\, G\, h^3} - \frac{6\, P\, z^3\, \nu}{3\, h^3\, E} - z\, \text{Constant2} + \text{Constant4};$$

Those expressions are now substituted into the earlier expressions for *u* and *w*:

$$u = \left(\int \epsilon_{xx}\, dx + Kz\right)[[1]];$$

$$w = \left(\int \epsilon_{zz}\, dz + Kx\right)[[1]];$$

We have four unknown constants, of which one is determined by the expression above for γ_{xz}. This means that three additional boundary conditions on the displacements are needed to prevent rigidbody displacement of the beam and thus determine a problem-specific solution. Two of them are perhaps obvious, namely to prevent displacement in the x and z directions at the neutral axis at the right-hand side support:

```
Condition2 = (u /. {x → L, z → 0}) == 0;
Condition3 = (w /. {x → L, z → 0}) == 0;
```

As explained in Figure 27 in the Timoshenko book the last condition may be to maintain a horizontal neutral axis at x=L and z=0:

```
Condition4basis = (∂x w /. {x → L, z → 0}) == 0;
```

... or to maintain a vertical cross-section tangent at that point:

```
Condition4alternative = (∂z u /. {x → L, z → 0}) == 0;
```

Now solving the four equations in the four unknowns:

```
solutionBasis =
  Solve[{Condition1, Condition2, Condition3, Condition4basis},
   {Constant1, Constant2, Constant3, Constant4}];
```

... which is also done for the alternative boundary condition at the fixed support:

```
solutionAlternative =
  Solve[{Condition1, Condition2, Condition3, Condition4alternative},
   {Constant1, Constant2, Constant3, Constant4}];
```

The symbolic expression for those solutions are:

```
uBasisSymbolic = (u /. solutionBasis)[[1]] // Expand
```

which yields: $-\frac{3\,P\,z}{2\,G\,h}+\frac{2\,P\,z^3}{G\,h^3}+\frac{6\,L^2\,P\,z}{h^3\,E}-\frac{6\,P\,x^2\,z}{h^3\,E}-\frac{2\,P\,z^3\,\nu}{h^3\,E}$

```
wBasisSymbolic = (w /. solutionBasis)[[1]]
```

which yields: $\frac{4\,L^3\,P}{h^3\,E}-\frac{6\,L^2\,P\,x}{h^3\,E}+\frac{2\,P\,x^3}{h^3\,E}+\frac{6\,P\,x\,z^2\,\nu}{h^3\,E}$

```
uAlternativeSymbolic = (u /. solutionAlternative)[[1]]
```

which yields: $\frac{2\,P\,z^3}{G\,h^3}+\frac{6\,L^2\,P\,z}{h^3\,E}-\frac{6\,P\,x^2\,z}{h^3\,E}-\frac{2\,P\,z^3\,\nu}{h^3\,E}$

```
wAlternativeSymbolic = (w /. solutionAlternative)[[1]] // Expand
```

which yields: $\frac{3 L P}{2 G h} - \frac{3 P x}{2 G h} + \frac{4 L^3 P}{h^3 E} - \frac{6 L^2 P x}{h^3 E} + \frac{2 P x^3}{h^3 E} + \frac{6 P x z^2 \nu}{h^3 E}$

For plotting purposes we also substitute values:

```
uBasis = uBasisSymbolic /. input2 /. input1;
wBasis = wBasisSymbolic /. input2 /. input1;

uAlternative = uAlternativeSymbolic /. input2 /. input1;
wAlternative = wAlternativeSymbolic /. input2 /. input1;
```

Comparison with Timoshenko expressions

For reference, the displacement expressions from the Timoshenko book are:

$$EI = E \frac{h^3}{12};$$

$$uTimoshenkoBasisSymbolic = -\frac{P z x^2}{2 EI} + \frac{2 P z^3}{G h^3} - \frac{P \nu z^3}{6 EI} + \left(\frac{P L^2}{2 EI} - \frac{3 P}{2 G h}\right) z \text{ // Expand;}$$

$$wTimoshenkoBasisSymbolic = \frac{P \nu x z^2}{2 EI} + \frac{P x^3}{6 EI} - \frac{P L^2}{2 EI} x + \frac{P L^3}{3 EI};$$

$$uTimoshenkoAlternativeSymbolic = -\frac{P z x^2}{2 EI} + \frac{2 P z^3}{G h^3} - \frac{P \nu z^3}{6 EI} + \frac{P L^2}{2 EI} z;$$

$$wTimoshenkoAlternativeSymbolic = \frac{P \nu x z^2}{2 EI} + \frac{P x^3}{6 EI} + \left(-\frac{3 P}{2 G h} - \frac{P L^2}{2 EI}\right) x + \frac{P L^3}{3 EI} + \frac{3 P L}{2 G h} \text{ // Expand;}$$

A comparison with the earlier expressions show they are the same:

```
uTimoshenkoBasisSymbolic - uBasisSymbolic
wTimoshenkoBasisSymbolic - wBasisSymbolic
uTimoshenkoAlternativeSymbolic - uAlternativeSymbolic
wTimoshenkoAlternativeSymbolic - wAlternativeSymbolic
```

which yields: 0

which yields: 0

which yields: 0

which yields: 0

Plot of neutral axis

```
Plot[{wBasis /. z → 0, wAlternative /. z → 0}, {x, 0, L /. input1},
 PlotStyle → {{Solid, Black}, {Dashed, Black}},
 PlotLegends → Placed[{"Horizontal NA", "Vertical cross-section"},
   Below]]
```



Comparison of displacement with Euler-Bernoulli and Timoshenko beam theory

(Notice that the differences can be very large for very short beams, which are not really beams at all but helpful for visualizing shear deformation.) The downwards tip deflection obtained above depends on the boundary condition assumed for the rotation at the fixed end:

```
wBasisSymbolic /. {x → 0, z → 0}
```

which yields: $\frac{4 L^3 P}{h^3 E}$

```
wBasisSymbolic /. {x → 0, z → 0} /. input2 /. input1
```

which yields: -675

```
wAlternativeSymbolic /. {x → 0, z → 0}
```

which yields: $\frac{3 L P}{2 G h} + \frac{4 L^3 P}{h^3 E}$

```
wAlternativeSymbolic /. {x → 0, z → 0} /. input2 /. input1
```

which yields: -967.5

In Euler-Bernoulli the tip deflection of a cantilever is

$$\text{wTipBernoulli} = \frac{P L^3}{3 EI}$$

which yields: $\frac{4 L^3 P}{h^3 E}$

```
wTipBernoulli /. input2 /. input1
```

which yields: -675

In Timoshenko beam theory, in which shear deformations are added to the Euler-Bernoulli flexural deformation, the tip displacement is:

$$\text{wTipTimoshenko} = \frac{P L}{G \frac{5}{6} h}$$

which yields: $\frac{6 L P}{5 G h}$

```
wTipTimoshenko /. input2 /. input1
```

which yields: $-234.$

That means that the total deflection of a Timoshenko beam is:

```
wTipBernoulli + wTipTimoshenko /. input2 /. input1
```

which yields: `-909.`

Plot of edges

Perhaps the most interesting outcome of determining the displacements in this example is to observe that plane sections do not remain plane in bending: